MAPLE E.D.O E MODELOS MATEMÁTICOS

AULA Nº 1.

INTRODUÇÃO

O Maple V é um software capaz de resolver problemas em diversos ramos da matemática. Dentro dos recursos disponíveis, temos o Cálculo Diferencial e Integral, bem como Gráficos de 2 e 3 dimensões. Este material tem como finalidade apresentar ao aluno conceitos iniciais deste aplicativo para servir de subsídio para os alunos que já estudaram Equações Diferenciais. Apresentaremos os principais comandos para resolução de equações diferenciais dando soluções exatas e gráficas. Daremos também, a resolução através do Maple, para alguns Modelos Matemáticos sejam eles: Decaimento Radioativo, Resfriamento de Newton, Crescimento Populacional, Taxas de Variação, Circuito Elétricos, Modelos Geométricos entre outros.

CONCEITOS INICIAIS

Uma linha de comando do MAPLE é caracterizada por > (prompt)

>

Ao final de cada linha de comando escrita no MAPLE devemos sempre colocar**:** (dois pontos) e teclar **ente,** neste caso o Maple executa o comando, ou colocar**;** (ponto e vírgula) e teclar **enter** e neste caso o Maple executa e exibe o resultado do comando.

Para as operações devem ser utilizados os símbolos:
+ (soma), - (diferença),* (produto),/ (divisão), ^(potência), x^(m/n)(para raiz n-ésima de x elevado a m-ésima potência), sqrt(x) (raiz quadrada de x), = (igual), <(menor),> (maior), <=(menor ou igual), >= (maior ou igual), <>(diferente).
O número **e** é representado por **exp(1)** e o número **pi** é representado por **Pi.**
Usa-se**:=** para se definir uma expressão.

PACOTES

Alguns comandos do Maple são de uso específico e por isto são agrupados em **pacotes** . Para disponibilizar estes comandos é preciso carregar o pacote que é feito com o comando ***with*** . Pode-se carregar mais de um pacote em uma mesma aplicação. Faremos uso nesta apostila dos seguintes pacotes:

> with(DEtools):
> with(plots):
> with(student):

Ou usando; no final de cada pacote que exibe o resultado.

COMANDOS MAIS USADOS PARA RESOLUÇÃO DE EQUAÇÕES DIFERENCIAIS

> dsolve(ode);
> dsolve(ode,y(x),extra_args);
> dsolve({ode,ICs},y(x),extra_args)
> dsolve({sysODE,ICs},{funcs},extra_args)

Os parâmetros tem os seguintes significados:

ode- uma equação diferencial ordinária
y(x)- qualquer função indeterminada de uma variável
ICs- condições iniciais de uma equação diferencial
{sysODE}- um conjunto com um sistema de equações diferenciais ordinárias
{funcs}-um conjunto com funções indeterminadas
extra_args- opcional, dependendo do tipo de problema a ser resolvido, como por exemplo:
"implicit"-quando se deseja a solução na forma implícita
"explicit"- quando se deseja a solução na forma explícita.

> **dsolve** é capaz de resolver diferentes tipos de problemas de E.D.O que incluem desde simples equações como sistemas de equações diferenciais, envolvendo ou não condições iniciais, bem como soluções numéricas dentre as quais estudaremos soluções por séries, alem de soluções usando transformadas de Laplace e de Fourier.

> DEplot(deqns, vars, trange, inits):

DEplot - este comando traça as curvas solução de uma EDO de primeira ordem ou de um sistema de equações de primeira ordem.
inits - condição inicial para o traçado de curvas solução num campo direcional.

OUTROS COMANDOS

rhs - Expressa o lado direito da expressão

Exemplo:

> **with(DEtools):**
> **e:=y=a*x^2+b;**

$$e := y = a\,x^2 + b$$

> **rhs(e);**

$$a\,x^2 + b$$

diff(f(x),x) - Encontra a 1ª derivada de uma função f(x).

Exemplo:

> **diff(cos(x),x);**

$$-\sin(x)$$

Diff(f(x),x) - visualiza a expressão.

Exemplo:
> **Diff(cos(x),x);**

$$\frac{\partial}{\partial x}\cos(x)$$

isolate - isola uma subexpressão no lado esquerdo da equação.

Exemplo:

> **readlib(isolate):**
> **isolate(x^2-3*x-5,x^2);**

$$x^2 = 3\,x + 5$$

> **isolate(4*x*sin(x)=3,sin(x));**

$$\sin(x) = \frac{3}{4}\frac{1}{x}$$

separablesol - Encontra a solução de uma EDO a variáveis separáveis de 1ª ordem.

Exemplo:

> **g:=t^2*(z(t)+1)+z(t)^2*(t-1)*diff(z(t),t)=0;**

$$g := t^2\,(z(t)+1) + z(t)^2\,(t-1)\left(\frac{\partial}{\partial t}z(t)\right) = 0$$

> **separablesol(g,z(t));**

$$\{\frac{1}{2}z(t)^2 - z(t) + \ln(z(t)+1) + \frac{1}{2}t^2 + t + \ln(t-1) = \_C_1\}$$

impliciplot - para traçado de gráfico na forma implícita.

Exemplo:

```
> with(plots):
> implicitplot3d(z=x^2+y^2,x=-3..3,y=-3..3,z=0..6);
```



OBSERVAÇÃO :

Se ao final dos comandos plot ou implicitplot colocarmos:
color=cor(altera a cor do gráfico para a cor especificada)
title= 'nome'(escreve o nome do gráfico)
scaling=constrained(coloca escalas iguais nos eixos horizontal e vertical)

Exemplo:

```
> with(plots):
> implicitplot(x^2+y^2=1,x=-3..3,y=-3..3,color=blue,title='modelo');
```



simplify - simplifica a expressão.
Exemplo:

> **simplify(4^(1/2)+3);**

$$5$$

> **simplify(sin(x)^2+cos(x)^2);**

$$1$$

unapply - expressa uma função de R em R.

Exemplo:

> **p:=x^2+sin(x)+1;**

$$p := x^2 + \sin(x) + 1$$

> **f:=unapply(p,x);**

$$f := x \rightarrow x^2 + \sin(x) + 1$$

> **q:=x^2+y^3+1;**

$$q := x^2 + y^3 + 1$$

> **t:=unapply(q,x);**

$$t := x \rightarrow x^2 + y^3 + 1$$

showtangent - gráfico da função e da reta tangente.

Exemplo:

> **with(student):**
> **showtangent(x^2+5,x=2);**



evalf - converte uma expressão numérica exata em um número decimal.

Exemplo:

> **evalf(Pi);**

$$3.141592654$$

> **int(exp(x^3),x=0..1);**

$$\int_0^1 e^{(x^3)}\,dx$$

> **evalf(%);**

$$1.341904418$$

fsolve - resolve uma expressão dando o resultado na forma decimal.

Exemplo:

> **q:=3*x^4-16*x^3-3*x^2+13*x+16;**

$$q := 3\,x^4 - 16\,x^3 - 3\,x^2 + 13\,x + 16$$

> **fsolve(q,x,1..2);**

$$1.324717957$$

display - executa uma lista de plotagem.

Exemplo:

> **F:=plot(cos(x),x=-Pi..Pi,y=-Pi..Pi,style=line):**
> **G:=plot(tan(x),x=-Pi..Pi,y=-Pi..Pi,style=point):**
> **display({F,G},axes=boxed,scaling=constrained,title='cosine and tangent');**

Veremos a seguir alguns exemplos que mostram como usar o comando ***dsolve*** na solução de simples equações diferenciais.

Exemplos:

Resolver as seguintes equações diferenciais ordinárias:

Exemplo 1:

> **g:=2*y(x)/(1-x^2);**

$$g := 2\,\frac{y(x)}{1-x^2}$$

> **ode:=diff(y(x),x)=g;**

$$ode := \frac{\partial}{\partial x}\,y(x) = 2\,\frac{y(x)}{1-x^2}$$

> **ans1:=dsolve(ode);**

$$ans1 := y(x) = \frac{\_C1\,x}{x-1} + \frac{\_C1}{x-1}$$

> **ic:=y(0)=y0;**

$$ic := y(0) = y0$$

> **soln:=dsolve({ode,ic},y(x));**

$$soln := y(x) = -\frac{y0\,x}{x-1} - \frac{y0}{x-1}$$

> **phi:=rhs(soln);**

$$\phi := -\frac{y0\,x}{x-1} - \frac{y0}{x-1}$$

> **plot({subs(y0=-1,phi),subs(y0=-3,phi)},x=1..10);**



Exemplo 2:

```
> h:=y(x)*ln(y(x))/sin(x);
```

$$h := \frac{y(x)\ln(y(x))}{\sin(x)}$$

```
> edo:=diff(y(x),x)=h;
```

$$edo := \frac{\partial}{\partial x}y(x) = \frac{y(x)\ln(y(x))}{\sin(x)}$$

```
> ans2:=dsolve(edo);
```

$$ans2 := y(x) = e^{(e^{-\_C1}\csc(x) - e^{-\_C1}\cot(x))}$$

```
> ic:=y(Pi/2)=exp(1);
```

$$ic := y\left(\frac{1}{2}\pi\right) = e$$

```
> soln:=dsolve({edo,ic},y(x));
```

$$soln := y(x) = e^{(e^{0}\csc(x) - e^{0}\cot(x))}$$

```
> cal:=rhs(soln);
```

$$cal := e^{(\csc(x) - \cot(x))}$$

```
> plot(cal,x=-1..1);
```



Exercício 1: Encontre a solução geral da equação $\frac{dy}{dx} = -\frac{x - y(x) - 1}{x + 4y(x) - 6}$

Exercício 2: Resolva a EDO $x + y'y = 0$, para $y(0) = y_0$ .Represente graficamente para $y_0 = 4$, $y_0 = 2$, para $-4 \le x \le 6$

Vejamos a seguir os comandos usados no traçado de campos direcionais de equações diferenciais e sistemas de equações diferenciais.

dfieldplot - traça campos direcionais

Exemplos:

Ex.1.

> **with(plots):**

> **with(DEtools):**

> **f:=-y/x;**

$$f := -\frac{y}{x}$$

> **ode:=diff(y(x),x)=f;**

$$ode := \frac{\partial}{\partial x} \mathrm{y}(x) = -\frac{y}{x}$$

> **dfieldplot(ode,y(x),x=-1..5,y=-1..6);**



Ex.2:

> **g:=-x/y;**

$$g := -\frac{x}{y}$$

> **ode1:=diff(y(x),x)=g;**

$$ode1 := \frac{\partial}{\partial x} \mathrm{y}(x) = -\frac{x}{y}$$

> **dfieldplot(ode1,y(x),x=-2..5,y=-2..5);**



Vamos traçar as curvas soluções:

> **inits:={[-2,0.5],[1,3]}:**
> **p1:=DEplot(ode1,y(x),x=-2..2,inits):**
> **p2:=dfieldplot(ode1,y(x),x=-2..2,y=-2..2):**
> **display({p1,p2});**



Exercício 1:

Trace o campo direcional e esboce as curvas soluções para $\frac{dy}{dx} = -y$.

Apresentaremos a seguir, as soluções, através do Maple, para alguns Modelos Matemáticos.

> **with(plots):**
> **with(student):**

> **with(DEtools):**

Exemplo 1.

Encontre a equação da curva sabendo-se que a inclinação da reta tangente num ponto qualquer é $\frac{-2x}{y}$.

> **ode1.1:=diff(y(x),x)=-2*x/y(x);**

$$ode11 := \frac{\partial}{\partial x}\mathrm{y}(x) = -2\,\frac{x}{\mathrm{y}(x)}$$

> **eq1:=separablesol(ode1.1,y(x));**

$$eq1 := \{\frac{1}{2}\mathrm{y}(x)^2 + x^2 = \_C_1\}$$

> **implicitplot({1/2*y^2+x^2 =1,1/2*y^2+x^2 =30,1/2*y^2+x^2 =3,1/2*y(x)^2+x^2 =8,1/2*y(x)^2+x^2 =13},x=-10..10,y=-10..10,color=blue,scaling=constrained);**



Exercício 1:

Determine a equação da curva sabendo-se que a 2ª derivada em cada ponto (x; y) é igual a $\mathbf{1/x^3}$. Sabe-se ainda que a curva passa pelo ponto (2 ; 4) e a equação da reta tangente neste ponto é **2y = 3x + 2**.

Exemplo 2:

A aceleração de uma partícula é diretamente proporcional ao tempo t. Para t = 0 seg. a velocidade da partícula é v = 9 m/s. Sabendo-se que ambas, velocidade e coordenada de posição são iguais a zero quando t = 3s, escreva a equação do movimento para a partícula.

> **eq3:=diff(v(t),t)=k*t;**

$$eq3 := \frac{\partial}{\partial t}\mathrm{v}(t) = k\,t$$

> **eq3.1:=dsolve({eq3,v(0)=9},v(t));**

$$eq31 := v(t) = \frac{1}{2} k\, t^2 + 9$$

> **solve(subs(v(t)=0,t=3,eq3.1),{k});**

$$\{k = -2\}$$

> **eq3.2:=subs(%,eq3.1);**

$$eq32 := v(t) = -t^2 + 9$$

> **phi:=rhs(eq3.2);**

$$\phi := -t^2 + 9$$

> **eq3.3:=diff(x(t),t)=phi;**

$$eq33 := \frac{\partial}{\partial t} x(t) = -t^2 + 9$$

> **eq3.4:=dsolve({eq3.3,x(0)=3},x(t));**

$$eq34 := x(t) = -\frac{1}{3} t^3 + 9\,t + 3$$

> **x(t):=rhs(%);**

$$x(t) := -\frac{1}{3} t^3 + 9\,t + 3$$

> **plot(x(t),t=-10..10,view=[-50..50,-50..50]);**



Exemplo 3**:**
Uma importante ferramenta na pesquisa arqueológica é a determinação da idade por rádio-carbono. Este é um modo de se determinar a idade de ossos humanos, plantas ou artefatos. O procedimento foi desenvolvido por W. Libby (1908-1980), no inicio dos anos 50 e lhe deu o premio Nobel de Química em 1960. A determinação de idade por rádio carbono está baseada no fato de que alguns restos (madeira, ossos, plantas) contêm quantidades residuais de carbono 14- C14, isótopo radioativo do carbono. Este isótopo é acumulado durante a vida e começa a decair com a sua morte. A meia vida de um isótopo radioativo significa o tempo em que a metade da quantidade original se decompõe. Como a meia vida do carbono

14 é longa, aproximadamente 5745 anos, quantidades mensuráveis de carbono 14 estão presentes após milhares de anos. Libby mostrou que, se aproximadamente 0,002 ou mais da quantidade original ainda estão presentes, então pode-se determinar precisamente a proporção da quantidade original de carbono 14 que resta, por dosagem de laboratório adequada.
Em outros termos:
Se Q(t) é a quantidade de C14 no tempo t e Q0 é a quantidade original, então a razão

Q(t)/Q0

poderá ser determinada, desde que esta quantidade não seja tão pequena.
a) Supondo que Q(t) satisfaça à equação dQ/dt = kQ, determine a constante k, de decaimento, para o carbono 14.
b) Encontre a expressão Q(t) em qualquer tempo t, se Q(0)= Q0.
c) Suponha que se descubram certos restos arqueológicos em que a quantidade residual de C14 seja 20% da quantidade original. Determine a idade desses restos.

Solução:

```
> eq5:=diff(Q(t),t)=k*Q(t);
```

$$eq5 := \frac{\partial}{\partial t} \mathrm{Q}(t) = k\, \mathrm{Q}(t)$$

```
> eq5.1:=dsolve(eq5,Q(t));
```

$$eq51 := \mathrm{Q}(t) = \_C1\, \mathrm{e}^{(k\, t)}$$

```
> eval(solve(subs(Q(t)=Q[0],t=0,eq5.1),{_C1}));
```

$$\{\_C1 = Q_0\}$$

```
> eq5.2:=subs(%,eq5.1);
```

$$eq52 := \mathrm{Q}(t) = Q_0\, \mathrm{e}^{(k\, t)}$$

```
> solve(subs(Q(t)=Q[0]/2,t=5745,eq5.2),{k});
```

$$\{k = -\frac{1}{5745} \ln(2)\}$$

```
> eq5.3:=subs(%,eq5.2);
```

$$eq53 := \mathrm{Q}(t) = Q_0\, \mathrm{e}^{\left(-\frac{1}{5745} \ln(2)\, t\right)}$$

```
> solve(subs(Q(t)=0.2*Q[0],eq5.3),{t});
```

$$\{t = 13339.47691\}$$

```
> soln:=rhs(eq5.3);
```

$$soln := Q_0\, \mathrm{e}^{\left(-\frac{1}{5745} \ln(2)\, t\right)}$$

>
plot({subs(Q[0]=6,soln),subs(Q[0]=.9,soln),subs(Q[0]=2,soln),subs(Q[0]=.5,soln),subs(Q[0]=-.6,soln),subs(Q[0]=1.5,soln),subs(Q[0]=3,soln)},t=-10000..10000);



Exercício 2:
Num castelo inglês existe uma velha mesa redonda de madeira que muitos afirmam ser a famosa Távola Redonda do rei Arthur. Por meio de um contador Geiger (instrumento que mede a radioatividade) constatou-se que a massa M existente na mesa, no ano de 1999, é 0,894 vezes a massa M0 de C!4 que existe num pedaço de madeira viva com o mesmo peso da mesa. M0 é também a massa de C!4 que existia na mesa quando esta foi feita há t anos. A mesa pode ser a famosa Távola Redonda?

Exemplo 4:
Uma certa cidade tinha uma população de 25.000 habitantes em1970 e, uma população de 30.000 em 1980. Assumindo que sua população continuará a crescer exponencialmente a uma taxa constante, que população podem seus planejadores urbanos esperar para o ano 2010?

> **eq7:=diff(P(t),t)=a*P(t);**

$$eq7 := \frac{\partial}{\partial t} \mathrm{P}(t) = a\, \mathrm{P}(t)$$

> **eq7.1:=dsolve({eq7,P(0)=25000},P(t));**

$$eq71 := \mathrm{P}(t) = 25000\, \mathbf{e}^{(a\, t)}$$

> **solve(subs(P(t)=30000,t=10,eq7.1),{a});**

$$\{a = \frac{1}{10} \ln\left(\frac{6}{5}\right)\}$$

> **eq7.2:=subs(%,eq7.1);**

$$eq72 := \mathrm{P}(t) = 25000\, \mathbf{e}^{\left(\frac{1}{10} \ln\left(\frac{6}{5}\right) t\right)}$$

> **subs(t=40,eq7.2);**

$$P(40) = 25000\, e^{\left(4 \ln\left(\frac{6}{5}\right)\right)}$$

> **evalf(%);**

$$P(40) = 51840.$$

> **P:=t->25000*exp(1/10*ln(6/5)*t);**

$$P := t \to 25000\, e^{\left(\frac{1}{10} \ln\left(\frac{6}{5}\right) t\right)}$$

> **plot(P(t),t=0..100);**



Exemplo 5:
Conhecemos de observações experimentais que a temperatura superficial de um objeto varia numa taxa proporcional à diferença entre a temperatura do objeto a do meio ambiente. Esta é a lei de resfriamento de Newton. Portanto, se T(t) é a temperatura do objeto no tempo t e Ta é a temperatura ambiente constante, temos a relação

$$\frac{dT}{dt} = k(T - T_a),\ k \in \Re$$

,
dependente do material de que é constituída a superfície do objeto. Usando estes dados, considere uma substância posta numa corrente de ar. Sendo a temperatura do ar 30° C e resfriando a substância de 100º C para 70º C em 15 minutos, encontre o momento em que a temperatura da substância será de 40º C.

> **eq10:=diff(T(t),t)=k1*(T(t)-30);**

$$eq10 := \frac{\partial}{\partial t} T(t) = k1\ (T(t) - 30)$$

> **eq10.1:=dsolve(eq10,T(t));**

$$eq101 := T(t) = 30 + e^{(k1\ t)}\ \_C1$$

> **solve(subs(T(t)=100,t=0,eq10.1),{_C1});**

$$\{\_C1 = 70 \frac{1}{e^0}\}$$

```
> evalf(%);
```

$$\{\_C1 = 70.\}$$

```
> eq10.2:=subs(%,eq10.1);
```

$$eq102 := T(t) = 30 + 70.\, e^{(k1\, t)}$$

```
> solve(subs(T(t)=70,t=15,eq10.2),{k1});
```

$$\{k1 = -.03730771920\}$$

```
> eq10.2:=subs(%,eq10.2);
```

$$eq102 := T(t) = 30 + 70.\, e^{(-.03730771920\, t)}$$

```
> solve(subs(T(t)=40,eq10.2),{t});
```

$$\{t = 52.15837877\}$$

```
> plot(30+70.*exp(-.3730771920e-1*t),t=-10..10);
```



Exercício 3:

O corpo de uma vítima de assassinato foi descoberto. O perito da polícia chegou à 1:00 h da madrugada e, imediatamente, tomou a temperatura do cadáver que era de 34,8º C. Uma hora mais tarde ele tomou novamente a temperatura e encontrou 34,1º C. a temperatura do quarto onde se encontrava a vítima era constante a 20º C. use a lei de resfriamento de Newton parar estimar a hora em que se deu a morte, admitindo que a temperatura normal de uma pessoa viva é de 36,5º C.

AULA Nº 2.

Apresentaremos a seguir, uma série de exercícios que devem ser resolvidos de maneira análoga aos exemplos vistos na aula anterior.

1) Determine a equação da curva tal que a inclinação da reta tangente em ponto qualquer (x;y) da curva é igual a metade da inclinação da reta que liga a origem ao ponto de tangência.

2) Um ator de cinema precisava fazer um regime para emagrecer, em virtude do seu papel num novo filme. O diretor exigiu que ele perdesse a terça parte do seu peso, que era de 120 kg, seguindo uma dieta que o emagrecesse proporcionalmente ao peso de cada instante. Nestas condições, sabendo-se que iniciada a dieta o artista emagreceu 20 kg em 40 dias, quanto tempo será necessário para que ele comece atuar no filme?

3) Um jarro de leite, inicialmente a 25ºC, é deixado para esfriar na varanda onde a temperatura é 0ºC. Suponha que a temperatura do leite tenha caído para 15ºC após 20 minutos. Quando a mesma será de 5ºC?

Circuito em série:

Em um circuito em série contendo apenas um resistor e um indutor, a segunda lei de Kirchhoff diz que a soma da queda de tensão no indutor (L(di/dt)) e da queda de tensão no resistor (iR) é igual à voltagem (E(t)) no circuito, ou seja

$L di/dt + Ri = E(t)$

4) Uma bateria de 12 volts é conectada a um circuito em série no qual a indutância é de 1/2 henry e a resistência, 10 ohms. Determine a corrente i se a corrente inicial é zero.

Circuito em série L-R-C:

A segunda lei de Kirchoff diz que a soma da queda de voltagem através do indutor, resistor e capacitor é igual à voltagem no circuito, ou seja :

$L\, di/dt + Ri + q/C = E(t)$

Como a carga q(t) no capacitor está relacionada com a corrente i(t) por i(t)=dq/dt temos:

$L\, dq/dt + R\, dq/dt + q/C = E(t)$

5) Encontre a carga q(t) no capacitor em um circuito em série L-R-C quando L=0,25 henry, R=10 ohms, C=0,001 farad, E(t)=0, q(0)=q coulombs e i(0)=0.

SOLUÇÃO DOS EXERCICIOS:

```
> with(plots):
> with(DEtools):
> with(student):
```

1)Determine a equação da curva tal que a inclinação da reta tangente em ponto qualquer (x;y) da curva é igual a metade da inclinação da reta que liga a origem ao ponto de tangência .

```
> z:=solve(subs(b=0,y(x)=x*a+b), a );
```

$$z := \frac{y(x)}{x}$$

```
> eq1:=diff(y(x),x)=z/2;
```

$$eq1 := \frac{\partial}{\partial x}y(x) = \frac{1}{2}\frac{y(x)}{x}$$

```
> eq1.2:=dsolve(eq1,y(x));
```

$$eq12 := y(x) = \_C1\sqrt{x}$$

```
> phi:=z=2;
```

$$\phi := \frac{y(x)}{x} = 2$$

```
> phi[1]:=rhs(isolate(phi,y(x)),x);
```

$$\phi_1 := 2\,x$$

```
> plot({9*sqrt(x),phi[1]},x=0..2);
```



2) Um ator de cinema precisava fazer um regime para emagrecer, em virtude do seu papel num novo filme. O diretor exigiu que ele perdesse a terça parte do seu peso, que era de 120 kg, seguindo uma dieta que o emagrecesse proporcionalmente ao peso de cada instante. Nestas condições, sabendo-se que iniciada a dieta o artista emagreceu 20 kg em 40 dias, quanto tempo será necessário para que ele comece atuar no filme?

```
> eq6:=diff(P(t),t)=r*P(t);
```

$$eq6 := \frac{\partial}{\partial t} \mathrm{P}(t) = r\, \mathrm{P}(t)$$

```
> eq6.1:=dsolve({eq6,P(0)=120},P(t));
```

$$eq61 := \mathrm{P}(t) = 120\, \mathrm{e}^{(r\, t)}$$

```
> solve(subs(P(t)=100,t=40,eq6.1),{r});
```

$$\{r = \frac{1}{40} \ln\left(\frac{5}{6}\right)\}$$

```
> eq6.2:=subs(r = 1/40*ln(5/6),eq6.1);
```

$$eq62 := \mathrm{P}(t) = 120\, \mathrm{e}^{\left(\frac{1}{40} \ln\left(\frac{5}{6}\right) t\right)}$$

```
> solve(subs(P(t)=80,eq6.2),{t});
```

$$\left\{ t = 40 \frac{\ln\left(\frac{2}{3}\right)}{\ln\left(\frac{5}{6}\right)} \right\}$$

```
> evalf(%);
```

$$\{t = 88.95604344\}$$

```
> plot(120*exp(1/40*ln(5/6)*t),t=-1000..1000);
```



3) Um jarro de leite, inicialmente a 25ºC, é deixado para esfriar na varanda onde a temperatura é 0ºC.Suponha que a temperatura do leite tenha caído para 15ºC após 20 minutos. Quando a mesma será de 5ºC ?

```
> eq10:=diff(T(t),t)=-h*(T(t)-T[a]);
```

$$eq10 := \frac{\partial}{\partial t} \mathrm{T}(t) = -h\,(\mathrm{T}(t) - T_a)$$

```
> eq10.1:=subs(T[a]=0,eq10);
```

$$eq101 := \frac{\partial}{\partial t} \mathrm{T}(t) = -h\,\mathrm{T}(t)$$

```
> eq10.2:=dsolve({eq10.1,T(0)=25},T(t));
```

$$eq102 := \mathrm{T}(t) = 25\,\mathbf{e}^{(-h\,t)}$$

```
> solve(subs(T(t)=15,t=20,eq10.2),{h});
```

$$\{h = -\frac{1}{20}\ln\left(\frac{3}{5}\right)\}$$

```
> eq10.3:=subs(h = -1/20*ln(3/5),eq10.2);
```

$$eq103 := \mathrm{T}(t) = 25\,\mathbf{e}^{\left(\frac{1}{20}\ln\left(\frac{3}{5}\right)t\right)}$$

```
> solve(subs(T(t)=5,eq10.3),{t});
```

$$\left\{t = -20\,\frac{\ln(5)}{\ln\left(\frac{3}{5}\right)}\right\}$$

```
> evalf(%);
```

$$\{t = 63.01320204\}$$

```
> T:=unapply(rhs(eq10.3),t);
```

$$T := t \to 25\,\mathbf{e}^{\left(\frac{1}{20}\ln\left(\frac{3}{5}\right)t\right)}$$

```
> plot(T(t),t=-100..100);
```



4) Uma bateria de 12 volts é conectada a um circuito em série no qual a indutância é de 1/2 henry e a resistência, 10 ohms. Determine a corrente i se a corrente inicial é zero.

> **c:=diff(i(t),t)+20*i(t)=24;**

$$c := \left(\frac{\partial}{\partial t}\mathrm{i}(t)\right) + 20\,\mathrm{i}(t) = 24$$

> **R:=dsolve(c,i(t));**

$$R := \mathrm{i}(t) = \frac{6}{5} + \mathbf{e}^{(-20\,t)}\,\_C1$$

> **s:=dsolve({c,i(0)=0},i(t));**

$$s := \mathrm{i}(t) = \frac{6}{5} - \frac{6}{5}\,\mathbf{e}^{(-20\,t)}$$

> **i(t):=rhs(s);**

$$\mathrm{i}(t) := \frac{6}{5} - \frac{6}{5}\,\mathbf{e}^{(-20\,t)}$$

> **plot(i(t),t=0..3);**



5) Encontre a carga q(t) no capacitor em um circuito em série L-R-C quando L=0,25 henry, R=10 ohms, C=0,001 farad, E(t)=0, q(0)=q coulombs e i(0)=0.

> **edo1:=(25/100)*diff(q(t),t$2)+10*diff(q(t),t)+1000*q(t)=0;**

$$edo1 := \frac{1}{4}\left(\frac{\partial^2}{\partial t^2}\mathrm{q}(t)\right) + 10\left(\frac{\partial}{\partial t}\mathrm{q}(t)\right) + 1000\,\mathrm{q}(t) = 0$$

> **sol:=dsolve(edo1,q(t));**

$$sol := \mathrm{q}(t) = \_C1\, e^{(-20\,t)}\cos(60\,t) + \_C2\, e^{(-20\,t)}\sin(60\,t)$$

> **soln:=dsolve({edo1,q(0)=q[0]},q(t));**

$$soln := \mathrm{q}(t) = q_0\, e^{(-20\,t)}\cos(60\,t) + \_C2\, e^{(-20\,t)}\sin(60\,t)$$

> **soln1:=dsolve({edo1,q(0)=q[0],D(q)(0)=0},q(t));**

$$soln1 := \mathrm{q}(t) = q_0\, e^{(-20\,t)}\cos(60\,t) + \frac{1}{3}q_0\, e^{(-20\,t)}\sin(60\,t)$$

UNIVERSIDADE FEDERAL DA BAHIA
INSTITUTO DE MATEMATICA
DEPARTAMENTO DE MATEMÁTICA
II BIENAL DA SBM

MINI-CURSO
MAPLE E.D.O

&



MODELOS



MATEMÁTICOS

AUTORAS:
MARIA CRISTINA MENEZES
MÁRCIA B. MENEZES

25 a 29 de **Outubro de 2004**